

## 第2章 東アジア地域・貿易リンクモデル構築に伴う プログラム解説

著者	植村 仁一
権利	Copyrights 日本貿易振興機構（ジェトロ）アジア 経済研究所 / Institute of Developing Economies, Japan External Trade Organization (IDE-JETRO) <a href="http://www.ide.go.jp">http://www.ide.go.jp</a>
シリーズタイトル	アジア経済研究所統計資料シリーズ
シリーズ番号	95
雑誌名	アジア長期経済成長のモデル分析（II）
ページ	35-64
発行年	2011
出版者	日本貿易振興機構アジア経済研究所 / Institute of Developing Economies (IDE-JETRO)
URL	<a href="http://hdl.handle.net/2344/00008881">http://hdl.handle.net/2344/00008881</a>

## 第2章

### 東アジア地域・貿易リンクモデル構築に伴うプログラム解説

植村 仁一

本章は、貿易リンクモデルの構築に伴うデータ処理・加工及び各国・地域の相手国別、財別輸入関数の推定のために作成した エクセルファイル上で動作する Visual Basic for Application (VBA) プログラムの解説を行うものである。これは、今年度を通じた作業に際して作成したプログラムの資料としての性格を兼ね、今後の貿易リンクモデル作成の参考資料を提供するものである。一般に貿易リンクモデルの推定は、国・財が複数あるために作業量が膨大なものになる傾向がある。そこで、これらプログラムの整備をしておくことにより、モデル開発作業が効率的に進められると期待される。

本章の構造は以下の通りである。まず、第1節で貿易リンクモデルの概要を解説し、リンクモデル構築に必要なデータ群と、そこから推定される輸入関数群について、必要とされる条件などを確認する。続く第2節では推定されるべき輸入関数群の共通構造について概観する。第3節では輸入関数群の推定に必要である「競争者価格」の定義と、実際の計算を行うためのプログラムの構造を解説する。第4～6節では、輸入関数の満たすべき符号条件と、それを満たすための定式化（ここでは、変数変換やラグ構造の様々な組合せを指す）を模索するための省力化プログラムについて解説している。

#### 第1節 貿易リンクモデルの概要

初めに、構築を目指す貿易リンクモデルを概観し、必要とされるデータとその定義について解説する。

貿易リンクモデルとは、一つ一つが単体で完結している各国モデルを、各国間の貿易構造を通じて接続することによって、全体を一つのマクロ計量モデルとみなすものである。


需要先決型の各国モデルでは、通常、「輸入価格」「輸出額」はモデル外から与えられる「外生変数」として扱われ、それを条件としてモデルを解き、均衡解としての「輸出価格」「輸入額」が決定される（もちろん同時にその他の内生変数も決定される）。ところで、複数の各国モデルを別々に取り扱う場合、ある国の輸出と別の国の輸入とは直接的な関係はない。一方貿易リンクモデルは、各国モデルで決定された輸入額を輸出国別に合算し、それを外生的に輸出国モデルに戻す、という手続きを踏むことにより、各国の輸出入に有機的な関連を持たせようとするものである。

また、一つの国を輸入主体として見たとき、通常の場合その国が価格決定者である

ことは少ない。従って輸入価格は各国モデルでは外生的に扱われる。しかし、貿易リンクの枠組みの中では、それは輸出国側の輸出価格（これは各国の事情である程度決まると考える）の「加重平均」のような形で決まる。そこで貿易リンクモデルは、各国の事情で決まる「輸出価格」「輸入額」を、相手国別に振り分けることにより、外生的に各国に与えられる「輸入価格」「輸出額」を発生させるシステム、といてよい。

リンクシステムと、それを取り巻く各国モデルの関係は以下の概念図で表すことができる。この図では、中国モデルに注目した場合の各変数の流れを示している。なお、章末附図1に概念図の詳細を提示してある。

図1. リンクシステムの概念図



図中、実線は「財」の流れを、点線は「価格」の流れを示す。

現在構築を目指している貿易リンクモデルでは、輸入関数を相手国別・財別に推定する。財の区分は一次産品：SITC 0, 1, 2, 4、原油：SITC 3、製造業品：SITC 5-9 とするが、原油についてはすべて外生扱いとし、輸入関数の推定は行わない。また、関数の定式化では、輸入の決定要因は、所得、輸入価格、輸出者の競争価格（第3節）とする。

## 第2節 輸入関数の定式化

上述の通り、貿易リンクモデルでは各国・地域の輸入関数を相手国別に推定してお

くことが必要となる。今回開発する貿易リンクモデルでは、説明変数として自国の所得水準（GDP）、輸入価格と国内価格の相対価格（もしくは輸入価格の絶対水準）に加え、相手国とその競争相手国群のオファー価格の相対価格を導入する。例えば一つの財について、ある国の輸出価格が変化した場合、輸入者は、その国の輸出価格と、その国の競争相手である国群の輸出価格とを比較し、より有利な条件で取引を行うということである。

国コード YYY の、XXX 国からの第 n 財の輸入関数の定式化は、

$$YYYMnXXX = f[YYYGDP, YYYPMLCn/YYYPGDP, XXXPXn/XXXPCnYYY ]$$

もしくは、輸入と国内の相対価格を用いない場合であれば、

$$YYYMnXXX = f[YYYGDP, YYYPMLCn, XXXPXn/XXXPCnYYY ]$$

となる。ここで、変数名の頭につけられた国コードはモデル報告国のデータであることを示し、変数名の末尾の国コードは対象国を示す（以下の説明では、国を特定しない場合や明らかである場合は省略している）。YYYMnXXX は実質化された YYY 国の XXX 国からの輸入額（ドル建て）、YYYGDP は YYY 国の GDP（不変価格、現モデルでは 2000 年価格）、YYYPMLCn は第 n 財の輸入価格（現地通貨建て）、YYYPGDP は国内一般物価を示し、同様に XXXPXn は XXX 国の輸出価格（対世界・ドル建て）である。XXXPCnYYY は YYY 国市場において XXX 国が直面する、「XXX 国の競争国群の輸出価格」を表す。この XXXPCnYYY は以下の式で定義される。

$$XXXPCnYYY = \sum_{k \neq XXX, YYY} \left( \frac{a_{k, YYY}^n}{1 - a_{XXX, YYY}^n} \right) PX_k^n$$

ここで、 $a_{k,j}^n$  は、第 n 財の j 国輸入市場における k 国のシェアを示す。従って、右辺括弧内は、YYY 国を輸入国とする市場における XXX 国以外の参加者全体を 1 とした時のシェア（ウェイト）を示すこととなる。、 $PX_k^n$  は k 国の輸出価格である。

なお、すべての k で世界全体をカバーしていれば、基準年におけるこの値は 1 となる。実際は世界全体をカバーしているわけではなく基準年の値は 1 と異なるため、運用上はすべての値を基準年値で除して調整する。

### 第 3 節 プログラム 1：競争者価格の算出

上述の「一国市場におけるある輸出者に対する競争者価格」は、各国の対世界輸出価格を、輸入国側の基準年の輸入額をウェイトとして加重平均を取ったものである。定義式自体は非常に単純なものであるが、実際の計算手続きは煩雑である。そこで、省力化のため、競争者価格を自動算出するプログラムを作成した。各国輸出価格に改

訂があった場合や入手可能時系列が伸びた場合、もしくは基準年の変更を行う場合など、日々（年々）のルーチン的な作業においても省力化の点で有用である。

### 3-1. ファイルの構造

エクセルファイルにはリンク参加国分の各国シート（15枚）があり、ここには財別・相手国別輸入額が格納されている。これとは別に各国の輸出価格が格納してあるシート（2枚：一次産品と製造業品）と、輸入シェア行列が格納してあるシート（2枚）がある。輸入シェア行列シートでは、各国データシートから基準年に当たる輸入額を抽出し、その年の輸入シェア行列を算出するようになっている。

さらに、競争者価格を格納するための空シートを2枚（PXC1及びPXC3シート）用意しておく。また、ファイル全体一枚目に、基準年入力と、参考用として計算の経過時間を書き出すシートを用意しておく。

### 3-2. 計算作業と結果の出力

元データをすべて格納してある「完全版」は表中で多くの再計算が行われるため、このファイル上でそのまま計算を行わせると速度が遅い。そこで、簡易版として国ごとに必要部分を複製した国別ファイルを用意する。具体的には輸入シェア行列シートに直接輸入シェア行列を入力しておくことで、15枚の各国データシートがすべて不要となる。実際の計算作業はそちらで行う（完全版でそのまま計算をした場合と比べ、数10倍の速度となる）。

一方完全版はデータの一元管理と最小限の計算を行うためのものとしておき、データ改訂などがあった場合に、シェア行列などを自動計算するところまで用いる。

計算結果はPXC1及びPXC3シートに出力される。報告国と相手国が一致する場合、競争者価格は当然存在しない。

### 3-3. VBA スクリプト

章末附表1に全体を掲載している。全体の構造は以下のようにになっている。

- (1) ComCode (1 or 3) による財の選択
- (2) 各国輸出価格読み込み
- (3) 輸入シェア行列の読み込み
- (4) 競争者価格の算出
- (5) 出力

プログラムそのものは上から下へと流れるだけのものである。ComCode=1で一次産品を、ComCode=3で製造業品を選択し、同一の手続きで計算が行われる。

## 第4節 輸入関数の符号条件

各国の財別・相手国別輸入関数群は、符号条件を厳密に満たしておかないと、それらを組み合わせて貿易リンクモデルとする際、(所得が増えると輸入が減るといような) 経済学におかしな挙動を取ってしまう懸念がある。実質輸入の推定式の係数に期待される符号条件は、所得については正、輸入価格（もしくはそれと国内価格の比）については負、輸出者と競争者のオファー価格比についても負である。すなわち、

$$\begin{matrix} (+) & & (-) & & & & (-) \\ \text{MnXXX} = f[ & \text{GDP,} & \text{PMLCn/PGDP,} & \text{XXXPXn/XXXPCn} & ] \end{matrix}$$

となる。これら輸入関数群はリンクモデルへの組み込みを主目的とするため、係数の統計的有意性よりも符号条件を重要視する。

輸入関数の推定に当たっては、変数の変換を様々に組合せ、符号条件を満たすものを探していくという「総当たり」で臨み、条件を満たすものの中から最終的にモデルに組み込む候補を絞り込んでいくというアプローチを行うこととする。変数変換は無変換を含めて4種類を想定する。

- (1) 無変換
- (2) 対数
- (3) 階差
- (4) 対数階差

また、特に階差変換を行う(3)と(4)については、期待する符号が得られない場合、説明変数として1期のラグ項を導入することにより、符号の逆転が期待される。つまり、

$$\begin{matrix} & & (+) & & & & (+) \\ \Delta \text{MnXXX} = f[ & \dots, & \Delta \text{PMLCn,} & \Delta (\text{XXXPXn/XXXPCn}) & ] \end{matrix}$$

という(期待されない)符号となっている場合、それは右辺と左辺の今期の階差同士が正の相関を持つことを意味する。ここで、説明変数を1期のラグ項で置き換えれば、

$$\begin{matrix} & & (-) & & & & (-) \\ \Delta \text{MnXXX} = f[ & \dots, & (\Delta \text{PMLCn}) (-1), & (\Delta (\text{XXXPXn/XXXPCn})) (-1) & ] \end{matrix}$$

を得られやすくなる(必ずそうなるというわけではない)。

## 第5節 輸入関数推定の省力化について

上述の変数変換とラグ項の導入を組み合わせると、ある国の、一つの財、一つの相手国からの輸入関数一本の推定に関して、左辺・被説明変数と右辺・各説明変数について、

(左辺)

無変換 対数 階差 対数階差

(右辺第1項：所得)

無変換 対数 階差 対数階差

(右辺第2項：輸入価格)

無変換 対数 階差 対数階差 無変換(1期ラグ) 対数(1期ラグ) 階差(1期ラグ) 対数階差(1期ラグ)

(右辺第3項：競争者とのオファー価格比)

無変換 対数 階差 対数階差 無変換(1期ラグ) 対数(1期ラグ) 階差(1期ラグ) 対数階差(1期ラグ)

だけの組合せが想定され、これだけで $4 \times 4 \times 8 \times 8 = 1024$ 通りとなる。これらの中には、「水準」(無変換)の左辺を所得の対数で説明する、といったあまり意味を持たない変数変換の組合せも存在している。実際のプログラム内ではそのような無意味な定式化についてはスキップし、左辺と右辺第1項(所得)については同一の変数変換をするという制約をつけている。

また、右辺第2項については輸入価格そのものか、国内との相対価格にするかという選択があり、これも勘案すると(上記制約を無視すれば)2048通りの組合せとなる。さらにそのそれぞれについて、被説明変数の1期ラグ項を説明変数として加えるかどうかの選択を加えると4096通りもの定式化を比較することとなる。その中から符号条件を満たし、かつ、統計的有意性のあるべく高いものを抽出するという作業が必要となる。

このように、1本の式あたり4096通りの組合せについて符号条件を満たし、有意性の高いものを選ぶところまでについてエクセルで自動的に行えるようにしたのが今年度の一つの成果である。以下、その概要と一部詳細部分について解説し、記録に残しておきたい。

## 第6節 プログラム2：輸入関数推定省力化

エクセルには「回帰分析」を行う関数が装備されている。今回作成したプログラムは、上記の被説明・説明変数の変数変換や回帰分析、符号条件の確認など、シート上でできることはなるべくシート上の関数を用い、繰り返し計算や結果の書き出し等、シート上では不可能な最低限と思われる部分だけをVBAスクリプトとして作成している。章末附表2に全体を掲載する。

### 6-1. ワークシートの構成

マクロ(VBA)付きエクセルファイルは、輸入側の国・地域(報告国)について一つ

ずつ用意する。シートの構成とそれぞれの主な機能は以下の通りである（番号は便宜的につけてある）。

#### (1) Data Center

報告国の基礎データ（GDP、GDP デフレーター、為替レート、財別ドル建て輸出入価格）、及び実質化された輸入（財別、相手国別）を保存しておく。また、相手国のデータとしては輸出物価（財別）、相手国が報告国市場で直面する競争者価格（財別）も合わせて保存しておく。このシートは米ドル建て輸出入価格から各国通貨ベースの輸出入価格の算出を行うほか、(2) DepVar 以降のシートが必要に応じ参照している。

#### (2) DepVar

被説明変数を抽出し、先に述べた変数変換を適用した系列を自動的に計算しておく。回帰分析を行うシートからの信号で、適当な系列を返す。具体的には、「無変換」「対数変換」「階差変換」「対数階差変換」の操作を行った列を順に並べておき、与えられた信号により DepVar シート内の別の列に必要な列を呼ぶ構造となっている。回帰分析を行うシートからは常にこの列を呼ぶことにより、構造が繁雑となることを防いでいる。

#### (3) GDP

#### (4) PM

#### (5) Rel. PM

#### (6) Rel. PXC

これら4つについても同様である（1期ラグ項も同様）。(4)は絶対輸入価格、(5)は相対輸入価格を作成しておくシートであり、この2つについては、回帰分析シートからの信号により、どちらか片方の情報のみが返される。(6)は輸出国側の輸出価格と、当該輸出国が報告国市場で直面する競争者の輸出価格との相対価格を計算するシートである。

#### (7) Data for OLS

回帰分析を行うためのデータを一旦保存。(9) Regression シートにはそのまま読み込まれるだけである。

#### (8) Base

報告国を指定（ファイルごとに固定）。また、相手国（1～15）、財コード（1 or 3）、相対輸入価格を使うか否か（使用=1、不使用=0）、説明変数の1期ラグを使うか否か（使用=1、不使用=0）を選択するシート。回帰係数のt値の目安の設定も行う。なお、実際の運用に当たっては一々手作業で選択せず、報告国以外はVBA スクリプト内で自動的に変化させて繰り返し計算を行う。



ところで、貿易データは相手国ごとに系列の開始年次が異なる場合があるため、回帰分析を行うためのデータ系列の長さをすべての相手国で同一にしておくことができない。このシートでは各相手国別のデータ開始年次を自動的に取得し、例外となる国があるかどうかを調べることも行っている。例外となる長さを持つ相手国については、次の (9) Regression シートにて分析期間を調整する (調整のみ手作業)。

#### (9) Regression

必要なデータセットから回帰分析を行い、結果を評価する。具体的には回帰分析を行うための LINEST(・) 関数を用いて係数と t 値を計算し、「符号条件を満たしているか」「設定した t 値の Critical Value を超えているか」を判断する。変数変換と説明変数のラグ項のすべての組合せ、すなわち、被説明変数の 1 期ラグ項を含めないケースで 2048 通り、含めるケースで 2048 通りの合計 4096 通りについて条件を満たすものがない場合、t 値の条件を 0.1 ずつゆるめて最初から計算をやり直す。t 値の条件は絶対値で 3.0 から始め、それが 0.0 となっても (符号条件の方で) 一つも期待される条件を満たすものが見つからない場合は計算を打ち切る。

期待される符号条件と、指定した大きさの t 値の Critical Value 以上であるか否かの判定を行い、すべて満たされる場合に "Good!"、一つでも満たさなければ "NG" の判定を下し、その結果は変数に格納しておき、後にその定式化を別シート (後述のシート 10 から 13) に書き出す。これを、14 の相手国、財 1 と財 3、輸入価格の絶対価格/相対価格別で次々に行っていく。

#### (補足) 有効なデータ系列の長さが異なる場合の処理について

相手国や財によって回帰分析に用いることのできるデータ系列の有効長が異なる場合がある。この場合、すべてを「短い方」に合わせるしてしまうのはデータを無駄にすることになるし、自由度確保の点からもふさわしくない。

特に中国やベトナムについては貿易データが十分に長く取ることができないため、この二国を相手国とする場合に有効なデータの長さが短くなる。全体で見ると、たいていの場合では「多くの国が同一の (長めの) データ長があり、一つか二つの (より短い系列しか使えない) 例外がある」ということが多い。そこで、このシートでは有効長の異なる系列に対する回帰分析結果を複数準備しておき、相手国や財別といった条件に従って参照先を変えるという方法を取る。

例として被説明変数のラグ項を説明変数に含まず、韓国と台湾、ベトナムではその他多くの国よりも短い (しかもベトナムの第 3 財のみはさらに短い) 系列しか使うことのできない中国について、その概要を記録しておく。

この例では、3 カ所にそれぞれデータ長の異なる回帰分析を行った結果を置いている。そして、横に立てられたフラグにより、最終的な計算結果として使うものを取り出す構造としている。すなわち、

- (イ) Majority に関する係数及び t 値 (1985-2009 : 分析結果)
- (ロ) ベトナムの第 1 財及び韓国、台湾に関する係数及び t 値 (1987-2009 : 分析結果)
- (ハ) ベトナムの第 3 財に関する係数及び t 値 (1988-2009 : 分析結果)

は常に算出されている。従ってこの場合、1985 年からのデータを使うように指定されている (イ) の欄は、(ロ) や (ハ) が最終結果として採用されるようなケース (1985 年や 1986 年には N/A 値が入っている) では回帰係数は算出されず、エラーとなっているが、そこをフラグで回避して拾わないようにしている。具体的には、セル F29 には「国コード=VNM かつ財コード=1」または「国コード=KOR」または「国コード=TWN」の 3 通りの時にフラグ 1 が立ち、セル F35 には「国コード=VNM かつ財コード=1」の場合のみフラグ 1 が立つ。セル F23 は  $1-(F29+F35)$  という式が入力されており、上記の場合以外すべてについてフラグ 1 が立つ。

LINEST (・) 関数の出力先はこれに応じて 3 通り  $\times$  2 (被説明変数の 1 期ラグ項を説明変数として導入するか否か) の 6 通り準備されており、それぞれ、上記  $3 \times 2 = 6$  通りの期間 (ラグ項入りの場合は 1 年分短くなる) について計算範囲を選択しておく。この部分のみが手作業で行うべき操作であり、計算と出力は自動で行われる。

#### (10) Out (A0)

条件を満たす定式化が見つかった場合の出力用シート。以下も同じ。

#### (11) Out (AW)

#### (12) Out (RO)

#### (13) Out (RW)

それぞれ、絶対価格・被説明変数のラグ項なし (Absolute/withOut)、相対価格・被説明変数のラグ項あり (Relative/With) などという意味である。

#### (14) Summary

相手国別、財別、絶対／相対価格別、被説明変数のラグ項の有無の別で、t 値がいくらの時に初めて条件を満たすものがいくつ見つかったかを書き出しておくシート。同時にそれぞれの相手国別でかかった時間も計算するようにしてある。

### 6-2. VBA スクリプト

章末附表 2 に全体を掲載しているが、ワークシート上では実行不可能であるか、可能であるとしても非効率である部分を VBA で自動処理するようにしている。全体の構造は以下のようになっている。繰り返し計算のループは 5 重である

- (1) 変数宣言部分
- (2) 報告国の設定
- (3) ループ

- (a) 相手国
  - (b) 輸入財
  - (c) 被説明変数のラグ項導入有無
  - (d) 絶対/相対輸入価格切り替え
  - (e) t 値 critical value の変更
- (4) 条件を満たす定式化候補の書き出し
- (5) サマリーシートへの情報書き出し

初めに、T\_Critical として判定に用いるための t 値の Critical Value を設定する（初期値は 3.0）。条件を満たすものがなければ、この値を 0.1 ずつ引き下げて計算を行う。最初から「候補」が多すぎる場合はこの値を 6.0 などに上げて良いが、反対に、ヒットするはずのものが低い t 値から始まる場合には時間がそれだけ浪費されることとなる。

また、刻み幅が 0.1 では多すぎる候補が出てしまう場合（例えば T\_Critical=1.3 では候補は一つもなかったが、T\_Critical=1.2 では数十の候補が出た、など）には、T\_Critical を小数第二位まで設定し、再計算することも可能にしてある。上の例の場合では、例えば T\_Critical=1.29 として計算を始めると、刻み幅が 0.01 となり、T\_Critical=1.2 となったところで計算を終わるようになる。実際にはそれより前のどこかで、より少ない候補を与える t 値が決まることを想定している。

#### 第7節 輸入関数推定の実際（中国の各相手国からの第1財輸入）

実際に、中国のデータを適用して輸入関数の推定を行った結果は次の通りである。財と相手国の組合せ、被説明変数のラグ項導入の有無ごとに、それぞれ 4096 通りの定式化で推定し、係数の有意性を t 値で判定している。t 値が 1 以上となったもののみ具体的な定式化を示す。ここでは中国の第1財輸入について例示する。全体の結果はデータ篇4にまとめて掲載している。表題の右にある数値は符号条件を満たした定式化全体の数を、左端の○、×はラグ項の有無を示し、その次の数字は条件を満たした場合の最大 t 値である。

ここでは右辺第2項を「輸入価格/国内価格」の相対価格で推定している。表には示していないが、この部分を輸入価格単体（絶対価格）でも推定するため、実際の計算量はさらに倍となる。

(例) 第1財・対日本 (×64、○46)

(ラグ項なしで 64 本、ラグ項ありで 46 本の式が符号条件を満たしている)

(×1.7)  $M1JPN = f[ GDP, \Delta \log(PM1LC(-1)/PGDP(-1)), \Delta \log(JPNPX1/JPNPC1) ]$

(○1.2)  $M1JPN = f[ GDP, \Delta \log(PM1LC(-1)/PGDP(-1)), \Delta \log(JPNPX1/JPNPC1), M1JPN(-1) ]$

(例) 第1財・対韓国 (×66、○62)

$$(\times 1.7) \log(M1KOR) = f[\log(GDP), PM1LC/PGDP, \Delta(KORPX1/KORPC1)]$$

$$(\times 1.7) \Delta(M1KOR) = f[\Delta(GDP), \Delta\log(PM1LC/PGDP), \Delta(KORPX1/KORPC1)]$$

$$(\bigcirc 2.2) M1KOR = f[GDP, \log(PM1LC/PGDP), \Delta(KORPX1/KORPC1), M1KOR(-1)]$$

$$(\bigcirc 2.2) M1KOR = f[GDP, \log(PM1LC/PGDP), \Delta\log(KORPX1/KORPC1), M1KOR(-1)]$$

(例) 第1財・対台湾 (×66、○16)

$$(\times 3.0) M1TWN = f[GDP, PM1LC/PGDP, TWNPX1/TWNPC1]$$

$$(\times 3.0) M1TWN = f[GDP, \log(PM1LC/PGDP), TWNPX1/TWNPC1]$$

$$(\times 3.0) M1TWN = f[GDP, \log(PM1LC/PGDP), \log(TWNPX1/TWNPC1)]$$

$$(\bigcirc 0.7) \log(M1TWN) = f[\log(GDP), PM1LC/PGDP, TWNPX1/TWNPC1, \log(M1TWN(-1))]$$

$$(\bigcirc 0.7) \log(M1TWN) = f[\log(GDP), PM1LC/PGDP, \log(TWNPX1/TWNPC1), \log(M1TWN(-1))]$$

(例) 第1財・対米国 (×79、○58)

$$(\times 2.0) M1USA = f[GDP, \Delta\log(PM1LC/PGDP), USAPX1/USAPC1]$$

$$(\times 2.0) M1USA = f[GDP, \Delta\log(PM1LC/PGDP), \log(USAPX1/USAPC1)]$$

$$(\bigcirc 1.1) M1USA = f[GDP, \Delta(PM1LC/PGDP), USAPX1/USAPC1, M1USA(-1)]$$

$$(\bigcirc 1.1) M1USA = f[GDP, \Delta(PM1LC/PGDP), \log(USAPX1/USAPC1), M1USA(-1)]$$

$$(\bigcirc 1.1) M1USA = f[GDP, \Delta(PM1LC/PGDP), \Delta(USAPX1/USAPC1), M1USA(-1)]$$

$$(\bigcirc 1.1) M1USA = f[GDP, \Delta(PM1LC/PGDP), \Delta\log(USAPX1/USAPC1), M1USA(-1)]$$

このように、ある程度機械的に輸入関数の「候補」を絞り込んでおけば、それらを実際にモデルに組み込み、モデル全体のパフォーマンスの優劣や安定性などを相互に比較することによって、最終的にモデルに用いる輸入関数を決定していく作業が容易になる。

#### 第8節 輸入関数推定・高速版の開発に向けて

上記プログラムは、各係数について符号条件を満たし、t 値がより高いものを抽出するために、t 値を徐々に変化（低下）させつつすべての条件を満たすものを抽出していた。高速化を目指した次バージョンでは、t 値を 3.0 から 0.1 刻みに 0.1 までの 30 段用意し、一つの定式化に対して 30 段分の係数算出と判定を並行的に行い、条件を満たすもののうち最大の t 値を取るものを抽出する。t 値のループが 1 段減ることになる分高速化されることが期待される。

【参考文献】

- [1] Sano Takao, O.Wakita and M.Kudo [1985] "Trade Link System," in *Economic Link System for ASEAN –ELSA– Final Report, Volume II*, Statistical Department, Tokyo; Institute of Developing Economies ed., pp.345-356.
- [2] Toida Mitsuru, IDE ELSA Group and IBM-TSC ELSA Group [1985] "The ELSA Link Model: Linkage of the Original Five ASEAN Countries, the East Asian NICs, the United States and Japan," in *The ELSA Link Model: Structure, Simulations, and Estimated Models*, IDE ELSA Group and IBM-TSC ELSA Group eds, Tokyo; Institute of Developing Economies, pp.1-74.
- [3] Toida Mitsuru and J.Uemura [2005] "Trade Link Method," in *FTAs in East Asia –Trade Link Model (I)-*, Toida and Uemura eds, Chiba; Institute of Developing Economies, JETRO, pp.447-482.
- [4] 野上裕生・植村仁一編[2011] 『アジア長期経済成長のモデル分析(I)』日本貿易振興機構アジア経済研究所。



【附表 1】 競争者価格の算出プログラム (VBA スクリプト)

Sub MakePXC()

```
' ----- Magic Numbers -----  
' 定数設定  
'  
' データ開始年、終了年および基準年の設定  
Const YearBegin = 1970, YearEnd = 2009, BaseYear = 2000  
'  
' リンク参加国数  
Const NCntry = 15 ' <<<<<===== Number of total countries  
  
' -----  
' 時系列の長さ (年)  
Dim Nyear As Byte  
Nyear = YearEnd - YearBegin + 1  
  
' -----  
' 参加国名コード  
Dim CountryName(NCntry) As String * 3  
  
'Share, PX シート内で↓この順に国が並んでいる必要あり  
CountryName(1) = "AUS"  
CountryName(2) = "CHN"  
CountryName(3) = "HKG"  
CountryName(4) = "IDN"  
CountryName(5) = "JPN"  
CountryName(6) = "KOR"  
CountryName(7) = "MYS"  
CountryName(8) = "NZL"  
CountryName(9) = "PHL"  
CountryName(10) = "SGP"  
CountryName(11) = "THA"  
CountryName(12) = "TWN"  
CountryName(13) = "USA"  
CountryName(14) = "VNM"  
CountryName(15) = "EUR"
```

```

' -----
'  輸入シェア行列の読み込み
'    << Read Share Matrix >>
' -----
'  一次産品 (Share1)、工業品 (Share2)
Dim ComCode As Byte
Dim Share(NCntry, NCntry, 2) As Double

Sheets("基準年入力").Select
Range("a7").Select
Selection.Value = "Data Read"
Range("a8").Select
Selection.Value = "  Begin: " & Now

For ComCode = 1 To 2 ' ComCode=1: Primary, ComCode=3: Manufacturing

Select Case ComCode
  Case 1
    Sheets("Share 0124").Select
  Case 2
    Sheets("Share 5-9").Select
End Select

  For zi = 1 To NCntry - 1          ' R.centry は EUR なし (別に定義)
    For zj = 1 To NCntry
      Range("a2").Select
      ActiveCell.Offset(zi, zj).Select
      Share(zi, zj, ComCode) = Selection.Value
    Next zj
  Next zi

' -----
'    << Read Export Price >>
' -----
Dim PX(YearEnd - YearBegin + 1, NCntry, 2) As Double

Dim BasePoint As Byte
Dim BasePrice As Double

BasePoint = BaseYear - YearBegin + 1

```



```

Select Case ComCode
    Case 1
        Sheets("PX1").Select
    Case 2
        Sheets("PX3").Select
End Select

For zi = 1 To Nyear
    For zj = 1 To NCntry
        Range("a1").Select
        ActiveCell.Offset(zi, zj).Select
        PX(zi, zj, ComCode) = Selection.Value
    Next zj
Next zi

For zj = 1 To NCntry
    BasePrice = PX(BasePoint, zj, ComCode)
    For zi = 1 To Nyear
        PX(zi, zj, ComCode) = PX(zi, zj, ComCode) / BasePrice
    Next zi
Next zj

Sheets("基準年入力").Select
Range("a9").Select
Selection.Value = "    End: " & Now

' -----
'      << Main Routine >>
' -----

Dim PXC(NCntry, YearEnd - YearBegin + 1, NCntry, 2) As Double

Dim SMtmp(NCntry, 2) As Double
Dim Denom(2) As Double
Dim TmpVal(2) As Double

Sheets("基準年入力").Select
Range("a10").Select
Selection.Value = "Calc."
Range("a11").Select

```

```

Selection.Value = "   Begin: " & Now

For i = 1 To NCntry           ' in the ith market
  For j = 1 To NCntry       ' reporting country
    For zj = 1 To NCntry   ' import share of PC
      SMtmp(zj, 1) = Share(i, zj, 1)
      SMtmp(zj, 2) = Share(i, zj, 2)
    Next zj
    Denom(1) = 1 - SMtmp(j, 1)
    Denom(2) = 1 - SMtmp(j, 2)
    ' EU については「自国」も「競争者」に入れる
    If j = 15 Then          ' except for EUR as RC
      ' すなわち、EU 以外について「自国」のシェアを 0 とする
      Else
        SMtmp(j, 1) = 0           ' force to be 0 for RC
        SMtmp(j, 2) = 0
      End If
    End If

  For t = 1 To Nyear
    TmpVal(1) = 0:   TmpVal(2) = 0
    For zj = 1 To NCntry
      For zk = 1 To 2
        TmpVal(zk) = TmpVal(zk) + (SMtmp(zj, zk) / Denom(zk)) * PX(t, zj, zk)
      Next zk
    Next zj
    If j = i Then
      PXC(i, t, j, 1) = 0
      PXC(i, t, j, 2) = 0
    Else
      PXC(i, t, j, 1) = TmpVal(1)
      PXC(i, t, j, 2) = TmpVal(2)
    End If
  Next t
Next j
Next i

Sheets("基準年入力").Select
Range("a12").Select
Selection.Value = "   End: " & Now

```

```

' -----
'      << Output >>
' -----

Sheets("基準年入力").Select
Range("a13").Select
Selection.Value = "Output"

Select Case ComCode
  Case 1
    Range("a14").Select
    Selection.Value = "   PXC1 Begin: " & Now
  Case 2
    Range("a16").Select
    Selection.Value = "   PXC3 Begin: " & Now
End Select

' 財別による出力先の変更
Select Case ComCode
  Case 1
    Sheets("PXC1").Select
  Case 2
    Sheets("PXC3").Select
End Select

GoSub HideAreas

For i = 1 To NCntry      ' in the ith market
  Range("a1").Select
  ActiveCell.Offset((i - 1) * (Nyear + 1), 0).Select
Selection.Value = CountryName(i)
  For j = 1 To NCntry   ' reporting cntry
    Range("a1").Select
    ActiveCell.Offset((i - 1) * (Nyear + 1), j). Select
Selection.Value = CountryName(j)
    For t = 1 To Nyear
      Range("a1").Select
      ActiveCell.Offset((i - 1) * (Nyear + 1) + t, 0).Select
      Selection.Value = YearBegin + (t - 1)
      Range("a1").Select
      ActiveCell.Offset((i - 1) * (Nyear + 1) + t, j).Select
    
```

```

' -----
                Selection.Value = PXC(i, t, j, ComCode)
' -----

                Next t
            Next j
        Next i

        GoSub RevealAreas

        Sheets("基準年入力").Select
        Select Case ComCode
            Case 1
                Range("a15").Select
                Selection.Value = "    PXC1  End:  " & Now
            Case 2
                Range("a17").Select
                Selection.Value = "    PXC3  End:  " & Now
        End Select

        GoTo SkipSubRoutine ' --- subroutine section ---
' -----
'     Subroutine Storage
' -----

        HideAreas:
            Rows("2:1000").EntireRow.Hidden = True
            Columns("b:p").EntireColumn.Hidden = True
        Return

        RevealAreas:
            Rows("2:1000").EntireRow.Hidden = False
            Columns("b:p").EntireColumn.Hidden = False
        Return
        SkipSubRoutine: ' --- end of subroutines ---
        Next ComCode

        Sheets("基準年入力").Select
        Range("a5").Select

        End Sub

```

## 【附表 2】 入力関数推定の省力化 (VBA スクリプト)

Excel VBA (Visual Basic for Application) は行番号によって管理する必要のない構造化言語であるが、ここに限り説明の便宜のために左端に行番号を附加している。

```
00010 Sub StartReg()
00020 ' 国名 : Reporting Country/ Partner Countries
00030 Dim RCountry As String
00040 Dim PCntryName(15) As String
00050
00060 ' 回帰分析を行う際の変数変換の条件格納変数
00070 ' それぞれ、Regression シートにおけるセル名としてある
00080 Dim H2, J2, K2, L2 As Byte
00090 Dim K3, L3 As Byte
00100
00110 ' 判定結果格納用。1 ループにつき 1024 通りなので配列は 2000 準備
00120 Dim Judgement(2000) As String
00130
00140 ' 定式化の変数名格納用
00150 Dim DepVar(2000) As String
00160 Dim IndepVar(2000, 4) As String
00170
00180 ' 諸条件をまとめて格納 (一時的な保存用)
00190 Dim Flags(2000) As String
00200
00210 ' 条件を満たした (良い) 結果の番号保存
00220 Dim GoodSpec(2000) As Boolean
00230
00240 ' カウンター変数
00250 Dim Counter As Integer
00260
00270 ' 条件を満たした (良い) 結果の数を数える
00280 Dim GoodCounter As Integer
00290
00300 ' 結果書き出しに際し、元のセル (列) への復帰用カウンター
00310 Dim ColSft As Byte
00320
00330 ' タイマー
00340 Dim BeginTime, Endtime As Variant
00350
```

```

00360 Dim SingleSpecFlag As Boolean
00370
00380 ' 満たすべき t 値の設定・再設定用
00390 Dim T_Critical, T_Increment As Double
00400 Dim T_CritForce As Double
00410
00420 ' t 値の初期設定値と処理終了用ストッパー値用
00430 Dim T_Init, T_Term As Double
00440
00450 ' 財コード、絶対/相対輸入価格、被説明変数ラグ項導入の有無 (2 値)
00460 Dim ComCode, RelativePM, WithWithoutLag As Byte
00470 Dim PCntryForce As String
00480
00490 ' 一国のみ再計算する場合の再設定用
00500 Dim ComCodeBegin, RelativePMBegin, WithWithoutLagBegin As Byte
00510 Dim ComCodeEnd, RelativePMEnd, WithWithoutLagEnd As Byte
00520 Dim PCntryBegin, PCntryEnd As Byte
00530
00540 Dim PCntrySelect(15) As Byte
00550 Dim zi As Byte
00560
00570 ' 結果を書き出すシート選択用スイッチ (4 種類の値を取る)
00580 Dim RelPMJump As Byte
00590
00600 ' 国番号保存用
00610 Dim QAUS, QCHN, QHKG, QIDN, QJPN, QKOR, QMYS, QNZL As Byte
00620 Dim QPHL, QSGP, QTHA, QTWN, QUSA, QVNM, QEUR As Byte
00630
00640 ' ループのカウンター及び一時格納用ダミー
00650 Dim i, ii As Integer
00660 Dim xxx As String ' dummy
00670
00680 ' 変数の初期化
00690 ' ----- init -----
00700 QAUS = 1: QCHN = 2: QHKG = 3: QIDN = 4: QJPN = 5: QKOR = 6
00710 QMYS = 7: QNZL = 8: QPHL = 9: QSGP = 10: QTHA = 11: QTWN = 12
00720 QUSA = 13: QVNM = 14: QEUR = 15
00730
00740 ' 変数名 (国名) の設定
00750 PCntryName(QAUS) = "AUS": PCntryName(QCHN) = "CHN"

```

```

00760 PCntryName(QHKG) = "HKG": PCntryName(QIDN) = "IDN"
00770 PCntryName(QJPN) = "JPN": PCntryName(QKOR) = "KOR"
00780 PCntryName(QMYS) = "MYS": PCntryName(QNZL) = "NZL"
00790 PCntryName(QPHL) = "PHL": PCntryName(QSGP) = "SGP"
00800 PCntryName(QTHA) = "THA": PCntryName(QTWN) = "TWN"
00810 PCntryName(QUSA) = "USA": PCntryName(QVNM) = "VNM"
00820 PCntryName(QEUR) = "EUR"
00830
00840 ' 報告国（自国）のワークシート上への設定
00850 Sheets("Base").Select
00860 RCountry = Range("a1").Value
00870
00880 ' Base シートにて一国のみの再計算を指定する場合
00890 PCntryForce = Range("l3").Value
00900 ComCodeBegin = Range("m3").Value
00910 RelativePMBegin = Range("n3").Value
00920 WithWithoutLagBegin = Range("o3").Value
00930 T_CritForce = Range("L6").Value
00940
00950 T_Init = 3# ' ----- initial value (T-Critical Value) ----- t 値の初期値
00960 SingleSpecFlag = False
00970 If PCntryForce = "" Then          '*** 「1 国のみ」の条件がない場合
00980     PCntryBegin = 1
00990     PCntryEnd = 15
01000     ComCodeBegin = 1
01010     RelativePMBegin = 0
01020     WithWithoutLagBegin = 0
01030     ComCodeEnd = 3
01040     RelativePMEnd = 1
01050     WithWithoutLagEnd = 1
01060     T_Increment = 0.1
01070     T_Term = 0
01080 Else                               '*** 「1 国のみ」の条件がある場合
01090     SingleSpecFlag = True
01100     For i = 1 To 15
01110         If PCntryForce = PCntryName(i) Then
01120             PCntryBegin = i
01130         Else
01140             End If
01150     Next i

```

```

01160         PCntryEnd = PCntryBegin
01170         ComCodeEnd = ComCodeBegin
01180         RelativePMEnd = RelativePMBegin
01190         WithWithoutLagEnd = WithWithoutLagBegin
01200         T_Increment = 0.1
01210         T_Init = T_CritForce
01220     End If
01230     T_Critical = T_Init
01240
01250
01260     *****
01270     ' 特定の国だけを対象とする場合の設定
01280     ' この場合は、下の zi ループで回し、ii ループを
01290     コメントアウト（1340 行と 4130 行）
01300     （全体で行う場合は 1320 行と 4140 行をコメントアウト）
01310     *****
01320     PCntrySelect(1) = QCHN: PCntrySelect(2) = QJPN
01330     PCntrySelect(3) = QKOR: PCntrySelect(4) = QTWN
01340     PCntrySelect(5) = QUSA
01350
01360     For zi = 2 To 3
01370         ii = PCntrySelect(zi)
01380         ' For ii = PCntryBegin To PCntryEnd ' （コメントアウト）
01390         ' 相手国別輸入関数の推定時、全部で 15 の国・地域があるが、
01400         ' その中で「自分自身」が相手国となった場合にはスキップ（4120 行へ）
01410         If PCntryName(ii) = RCountry Then GoTo SkipThisCountry
01420
01430         Sheets("Base").Select
01440         Range("b3").Select
01450         Selection.Value = PCntryName(ii)
01460         PCntry = Selection.Value
01470
01480         Range("b7").Value = T_Critical
01490
01500         ' 計算開始時刻設定
01510         BeginTime = Time
01520         *****
01530         ' ----- ComCode Loop          (ループ 2/5) -----
01540         ' 財コード「1」「3」を発生
01550         For ComCode = ComCodeBegin To ComCodeEnd Step 2

```



```

01560 Sheets("Base").Select
01570 Range("d3").Select
01580 Selection.Value = ComCode
01590
01600 ' ----- With/Withouth Lag Loop (ループ 3/5) -----
01610 ' 被説明変数のラグ項を説明変数に含めるか
01620 For WithWithoutLag = WithWithoutLagBegin To WithWithoutLagEnd
01630 Sheets("Base").Select
01640 Range("h3").Select
01650 Selection.Value = WithWithoutLag
01660
01670 ' ----- Relative PM Loop (ループ 4/5) -----
01680 ' 絶対/相対輸入価格を用いるか
01690 For RelativePM = 1 To 1 ' RelativePMBegin To RelativePMEnd
01700 Sheets("Base").Select
01710 Range("f3").Select
01720 Selection.Value = RelativePM
01730 '
01740
01750 T_Critical = T_Init ' refresh t-critical value
01760 ' ----- increment change -----
01770 ' 判定用 t 値の初期設定値 (通常開始時は 3.0)
01780 ' 初期設定値が小数第二位まで設定されていた場合、そこから
01790 ' 刻み幅 0.01 で切りの良い数字まで下げる。例えば T_Critical = 1.39 と
01800 ' 設定すると、T_Term = 1.3 までで計算を打ち切る。
01810 '
01820 Sheets("Base").Select
01830 If T_Critical * 100 - Int(T_Critical * 10) * 10 <> 0 Then
01840     T_Increment = 0.01
01850     T_Term = Int(T_Critical * 10) / 10 - 0.01 ' --- Stopper ---
01860 Else
01870     T_Increment = 0.1
01880     T_Term = 0
01890 End If
01900
01910 Sheets("Regression").Select
01920
01930 Range("d2").Select
01940 Selection.Value = T_Term
01950

```

```

01960 GoodCounter = 0
01970 ' ループ 5/5
01980 ' -----
01990 ' ----- メインルーチン (ここから) -----
02000 ' -----
02010 Do While GoodCounter = 0 ' ----- regression loop -----
02020 Sheets("Base").Select
02030 Range("b7").Select
02040 Selection.Value = T_Critical
02050
02060 Sheets("Regression").Select
02070
02080 Counter = 0
02090 GoodCounter = 0
02100 ' ----- 変数変換 -----
02110 ' --- 被説明変数 ---
02120 For H2 = 0 To 3 ' サブループ(5) 1/6
02130     Range("h2").Select
02140     Selection.Value = H2 ' DepVar Transformation
02150     For J2 = 0 To 3 ' サブループ(5) 2/6
02160     ' GDP は被説明変数と同じ変数変換とする。それ以外はスキップ(2600行へ)
02170     If J2 <> H2 Then GoTo SkipThisTransformation
02180         Range("j2").Select
02190         Selection.Value = J2 ' Income
02200         For K2 = 0 To 3 ' サブループ(5) 3/6
02210             Range("k2").Select
02220             Selection.Value = K2 ' PM
02230             For L2 = 0 To 3 ' サブループ(5) 4/6
02240                 Range("l2").Select
02250                 Selection.Value = L2 ' PX/PXC
02260                 ' ----- 1期ラグ項の使用 -----
02270                 ' --- 輸入価格 ---
02280                 For K3 = 0 To 1 ' サブループ(5) 5/6
02290                     Range("k3").Select
02300                     Selection.Value = K3
02310                     ' --- オファー価格と競争者価格比 ---
02320                     For L3 = 0 To 1 ' サブループ(5) 6/6
02330                         Range("l3").Select
02340                         Selection.Value = L3
02350

```

```

02360         Counter = Counter + 1
02370         GoodSpec(Counter) = False
02380         Judgement(Counter) = Range("e3").Value
02390             If Range("e3") = "Good!" Then
02400                 GoodSpec(Counter) = True
02410                 GoodCounter = GoodCounter + 1
02420                 DepVar(Counter) = Range("h4").Value
02430                 IndepVar(Counter, 1) = Range("j4").Value
02440                 IndepVar(Counter, 2) = Range("k4").Value
02450                 IndepVar(Counter, 3) = Range("l4").Value
02460                 IndepVar(Counter, 4) = Range("m4").Value
02470             Else
02480             End If
02490         Range("f3").Value = GoodCounter
02500         ' 諸条件をまとめて格納
02510         Flags(Counter) = H2
02520         Flags(Counter) = Flags(Counter) & " " & J2
02530         Flags(Counter) = Flags(Counter) & " " & K2
02540         Flags(Counter) = Flags(Counter) & " " & L2
02550         Flags(Counter) = Flags(Counter) & " / " & K3 & " " & L3
02560         Next L3             ' (5) 6/6 終わり
02570         Next K3             ' (5) 5/6 終わり
02580         Next L2             ' (5) 4/6 終わり
02590         Next K2             ' (5) 3/6 終わり
02600 SkipThisTransformation: ' GDP が被説明変数と違う変換の場合のスキップ先
02610     Next J2                 ' (5) 2/6 終わり
02620 Next H2                     ' (5) 1/6 終わり
02630
02640 ' ----- T-Critical Value Reset -----
02650 ' 条件を満たすものがなければ t 値の Critical Value を決められた
02660 ' 刻みで引き下げ、新たなループに
02670 T_Critical = T_Critical - T_Increment
02680
02690 ' その過程で Critical Value が設定した終了値となったら (符号条件を
02700 ' 満たすものが一つもなければ) この関数の推計自体をスキップ
02710 If T_Critical < T_Term Then GoTo SkipThisSpec
02720
02730 Loop ' ----- regression loop -----
02740 ' -----
02750 ' ----- メインルーチン (ここまで) -----

```

```

02760 '-----
02770 '----- 一つの定式化に対する推定 (ループ 5/5 : 終端) -----
02780 SkipThisSpec: '=== 符号条件を満たす定式化がない場合のスキップ先 ===
02790
02800 '----- output -----
02810 ' 絶対/相対輸入価格及び被説明変数のラグ項の有無による
02820 ' スイッチ変数の設定 (0~3)
02830 RelPMJump = 2 * RelativePM + WithWithoutLag
02840 Select Case RelPMJump
02850     Case 0
02860         Sheets("Out(AO)").Select
02870     Case 1
02880         Sheets("Out(AW)").Select
02890     Case 2
02900         Sheets("Out(RO)").Select
02910     Case 3
02920         Sheets("Out(RW)").Select
02930     Case Else
02940         GoTo BreakFromHere           ' ストッパー
02950     End Select
02960
02970 ' 書き出しの間は、行を非表示 (時間の短縮化のため)
02980 Rows("3:20000").EntireRow.Hidden = True
02990
03000 ' 選択された出力用シートの最下端を探す
03010 Range("a3").Select
03020 xxx = Selection.Value
03030 Do Until xxx = ""
03040     ActiveCell.Offset(1, 0).Select
03050     xxx = Selection.Value
03060 Loop
03070
03080 ' 書き出し
03090 If GoodCounter <> 0 Then
03100     For i = 1 To Counter
03110         ColSft = 0
03120         If GoodSpec(i) = True Then
03130             Selection.Value = i
03140             ActiveCell.Offset(0, 1).Select
03150             ColSft = ColSft + 1

```

```

03160      Selection.Value = Flags(i)
03170          ActiveCell.Offset(0, 1).Select
03180          ColSft = ColSft + 1
03190      Selection.Value = Judgement(i)
03200          ActiveCell.Offset(0, 1).Select
03210          ColSft = ColSft + 1
03220      Selection.Value = DepVar(i)
03230          ActiveCell.Offset(0, 1).Select
03240          ColSft = ColSft + 1
03250      Selection.Value = IndepVar(i, 1)
03260          ActiveCell.Offset(0, 1).Select
03270          ColSft = ColSft + 1
03280      Selection.Value = IndepVar(i, 2)
03290          ActiveCell.Offset(0, 1).Select
03300          ColSft = ColSft + 1
03310      Selection.Value = IndepVar(i, 3)
03320      If WithoutLag = 1 Then
03330          ActiveCell.Offset(0, 1).Select
03340          ColSft = ColSft + 1
03350          Selection.Value = IndepVar(i, 4)
03360      Else
03370      End If
03380      ActiveCell.Offset(1, -ColSft).Select
03390      Else
03400      End If
03410
03420      Next i
03430      Else
03440          Sheets("Base").Select
03450      End If
03460
03470      ' 行の再表示
03480      Rows("1:20000").EntireRow.Hidden = False
03490
03500      ' ----- Summary Statistics -----
03510      ' サマリーシートへの情報書き出し (1 / 2)
03520      Sheets("Summary").Select
03530
03540      ' 最下端を探す
03550      Range("a1").Select

```

```

03560 xxx = Selection.Value
03570 Do Until xxx = ""
03580     ActiveCell.Offset(1, 0).Select
03590     xxx = Selection.Value
03600 Loop
03610
03620 ' 相手国、財、輸入価格、ラグ項有無
03630 Selection.Value = PCntry
03640 ActiveCell.Offset(0, 1).Select
03650 Selection.Value = ComCode
03660 ActiveCell.Offset(0, 1).Select
03670 Selection.Value = RelativePM
03680 ActiveCell.Offset(0, 1).Select
03690 Selection.Value = WithWithoutLag
03700
03710 ' t 値 Critical Value
03720 ActiveCell.Offset(0, 1).Select
03730 Selection.Value = T_Critical + T_Increment
03740
03750 ' 定式化番号
03760 ActiveCell.Offset(0, 1).Select
03770 Selection.Value = GoodCounter
03780
03790 ' Regression シートに移動
03800 Sheets("Regression").Select
03810 Next RelativePM ' --- 絶対/相対輸入価格 (ループ 4/5 : 終端) -----
03820 ' ----- Relative PM Loop -----
03830 Next WithWithoutLag ' ----- ラグ項有/無 (ループ 3/5 : 終端) -----
03840 ' ----- With/Wighout Lag Loop -----
03850 Next ComCode ' ----- 次の財へ (ループ 2/5 : 終端) -----
03860 ' ----- ComCode Loop -----
03870
03880 BreakFromHere: ' ===== エラー・緊急時スキップ先 =====
03890
03900 ' 計算終了時刻設定
03910 Endtime = Time
03920
03930 If SingleSpecFlag = True Then ' 国・財条件付きの時は時間表示しない
03940     Else ' 無条件の場合
03950         ' サマリーシートへの情報書き出し (2 / 2)

```

```

03960      Sheets("Summary").Select
03970      ' 最下端を探す
03980      Range("a1").Select
03990      xxx = Selection.Value
04000      Do Until xxx = ""
04010          ActiveCell.Offset(1, 0).Select
04020          xxx = Selection.Value
04030      Loop
04040
04050      Selection.Value = PCntry
04060      ActiveCell.Offset(0, 1).Select
04070      Selection.Value = BeginTime
04080      ActiveCell.Offset(0, 1).Select
04090      Selection.Value = Endtime
04100  End If
04110
04120  SkipThisCountry:  '==== 相手国が自分自身の場合のスキップ先 ====
04130  ' Next ii '----- 次の相手国へ(ループ 1/5:終端) ----- (コメントアウト)
04140  Next zi '----- Selected Country Loop -----
04150
04160  ' カーソルを先頭に復帰
04170  Range("a2").Select
04180
04190  ThisWorkbook.Save
04200
04210  EndofThisMacro:  '===== 緊急避難用ジャンプ先 =====
04220
04230  End Sub

```